

# Лабораторная работа: нахождение суммы ряда

Использование оператора цикла  
с предусловием

## Постановка задачи

Даны действительные числа  $x$ ,  $\text{eps}$  ( $|x| < 1$ ;  $\text{eps} > 0$ ). Вычислить значение функции  $f(x)$  для данного  $x$  с помощью разложения этой функции в ряд с точностью  $\text{eps}$ .

Решим для функции  $\sin x$ :

$$f(x) = \sin x = \sum_{i=1}^{\infty} (-1)^{i+1} \frac{x^{2i-1}}{(2i-1)!}$$

$$f(x) = \sin x = \sum_{i=1}^{\infty} (-1)^{i+1} \frac{x^{2i-1}}{(2i-1)!}$$

Найдем несколько первых слагаемых  
ряда:

$$f(x) = \frac{x^1}{1!} - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \frac{x^9}{9!} - \square + (-1)^{2i-1} \frac{x^{2i-1}}{(2i-1)!}$$

## Основы программирования/Pascal

Например, для  $x = 0,5$  значение первых пяти слагаемых ряда равно

$$f(0.5) = \frac{0.5^1}{1!} - \frac{0.5^3}{3!} + \frac{0.5^5}{5!} - \frac{0.5^7}{7!} + \frac{0.5^9}{9!} - \square =$$

$$= \frac{0.5}{1} - \frac{0.5^3}{1 \cdot 2 \cdot 3} + \frac{0.5^5}{1 \cdot 2 \cdot 3 \cdot 4 \cdot 5} - \frac{0.5^7}{1 \cdot 2 \cdot 3 \cdot 4 \cdot 5 \cdot 6 \cdot 7} +$$

$$+ \frac{0.5^9}{1 \cdot 2 \cdot 3 \cdot 4 \cdot 5 \cdot 6 \cdot 7 \cdot 8 \cdot 9} - \square =$$

$$= \frac{0.5}{1} - \frac{0.125}{6} + \frac{0.03125}{120} - \frac{0.0078125}{5040} + \frac{0.001953125}{362880} - \square =$$

## Основы программирования/Pascal

$$\begin{aligned} &= 0.5 - \\ &- 0,020833333333333333 + \\ &+ 0,0002604166666666667 - \\ &- 0.00000155009920634921 + \\ &+ 0.00000000538228891093474 - \dots \end{aligned}$$

Видно, что вклад каждого последующего слагаемого существенно меньше предыдущего. То есть по модулю они уменьшаются.

Требуемая точность считается достигнутой, если модуль очередного слагаемого станет меньше заданного близкого к нулю числа  $\epsilon$ .

Например, если  $\epsilon=0.000001$ , то уже на пятом слагаемом суммирование можно прекратить, а найденную сумму объявить ответом.

Для нахождения суммы требуется  
последовательно находить слагаемые  
до тех пор, пока очередное из них не  
станет по модулю меньше  $\epsilon$ :

$$|a_i(x)| < \epsilon$$

Для нахождения слагаемых можно воспользоваться формулой

$$a_i(x) = (-1)^{i+1} \frac{x^{2i-1}}{(2i-1)!}$$

но проще и быстрее использовать рекуррентное соотношение, позволяющее находить каждое новое слагаемое через предыдущее:

$$\begin{cases} a_1 - \text{дано} \\ a_i(x) = a_{i-1}(x) \cdot g_i(x) \end{cases}$$

Выведем рекуррентное соотношение.

Первое слагаемое находим по формуле.

У нас оно равно

$$a_1(x) = x$$



Формула для каждого слагаемого:

$$a_i(x) = (-1)^{i+1} \frac{x^{2i-1}}{(2i-1)!}$$

Подставим вместо индекса  $i$  индекс  $i-1$ :

$$a_{i-1}(x) = (-1)^{i-1+1} \frac{x^{2(i-1)-1}}{(2(i-1)-1)!} = (-1)^i \frac{x^{2i-3}}{(2i-3)!}$$

Найдем отношение  $a_i(x) a_{i-1}(x)$ :

$$\frac{a_i(x)}{a_{i-1}(x)} = \frac{(-1)^{i-1} \frac{x^{2i-1}}{(2i-1)!}}{(-1)^i \frac{x^{2i-3}}{(2i-3)!}} = - \frac{x^2}{(2i-2)(2i-1)}$$

## Основы программирования/Pascal

Таким образом, рекуррентное соотношение будет иметь вид:

$$\begin{cases} a_1(x) = x \\ a_i(x) = a_{i-1}(x) \times \left(-\frac{x^2}{(2i-2)(2i-1)}\right), \text{ для } i = 2, 3, 4, 5, \dots \end{cases}$$

Используем это соотношение в программе:

# Основы программирования/Pascal

[Программа вычисления значения функции разложением в ряд](#)

[\(часть 1\)](#)  
program ExSin;

uses Crt;

const

imax = 10000; {максимальное  
количество слагаемых}

var

x, x2 : real;

s, a : real;

EPS : real;

i : word;

# Основы программирования/Pascal

## Программа вычисления значения функции разложением в ряд

### (часть 2):

```
begin
  ClrScr;

  {Ввод}
  writeln('Нахождение значения функции  $F(X)=\sin(X)$  ');
  writeln('через ее разложение в ряд. ');

  repeat
    write('Введите аргумент X ( $|X|<1$ ): ');
    readln(X);
  until (abs(X)<1);

  repeat
    write('Введите точность EPS ( $EPS>0$ ): ');
    readln(EPS);
  until (EPS>0);
```

# Основы программирования/Pascal

## Программа вычисления значения функции разложением в ряд

(часть 3):

```
{Нахождение суммы ряда с точностью eps}
a:=X;           {Очередное слагаемое}
i:=1;          {Индекс слагаемого}
s:=a;          {Сумма ряда. Начинаем с a1}
X2:=sqr(X);    {Квадрат X. Для вычислений.}

while (abs(a)>EPS) and (i<=imax) do
    {пока не достигнута точность}
begin
    inc(i);     {Увеличиваем индекс на 1}
    a:=a*(-X2/(2*i-2)/(2*i-1)); {Находим
следующее слагаемое через предыдущее}
    s:=s+a;    {Прибавляем слагаемое к сумме}
end;
```

# Основы программирования/Pascal

## Программа вычисления значения функции разложением в ряд

```
{Вывод ответа}
(часть 4).
writeln;
if i<=imax then
begin
  writeln('Значение функции F(x)=sin x');
  writeln('F(X)=' ,s:14:12,' -с помощью ряда;');
  writeln('F(X)=' ,sin(X):14:12,' -станд.функция');
end
else
  writeln('не найдено. Ряд расходится.');
```

  

```
{Задержка до нажатия на любую клавишу}
writeln('Нажмите любую клавишу...');
repeat
until Keypressed;
end.
```

# Основы программирования/Pascal

Для тестирования работы программы  
используем систему тестов:

Номер теста	Вход		Выход
	x	eps	
1	0.5	1E-3	0.4794270833 0.4794253860
2	0.99	1E-6	0.8360259784 0.8360259786
3	0.1	1E-2	0.0998333333 0.0998334166

По итогам работы подготовить **отчет**:

- Титульный лист
- Постановка задачи
- Метод решения
- Вывод рекуррентного соотношения
- Текст программы (с комментариями)
- Таблица тестов
- Вывод